RTC-AD12E
RTC ADC-DAC Expansion Board
Technical Manual

Release 1.0
10/95

Micromint, Inc.
902 Waterway Place
Longwood, FL 32750
www.micromint.com

Table of Contents

COPYRIGHT

Micromint, Inc.
902 Waterway Place
Longwood, FL 32750

DISCLAIMER & TERMS OF SALE

While we have attempted to provide accurate and up-to-date information in this manual, Micromint Inc. makes no representations or warranties respecting its contents. We reserve the right to make periodic changes to the text and to issue new editions of this manual without notification.

Occasionally in this manual we refer to other manufacturers' products. Such references do not constitute an endorsement of these products, but are included for the purpose of illustration or clarification. We do not intend such technical information and interface data to supersede information provided by individual manufacturers.

Devices sold by Micromint are covered by the warranty and patent indemnification provisions appearing in its Terms of Sale only. Micromint makes no warranty, express, statutory, implied, or by description regarding the information set forth herein or regarding the free merchantability or fitness for any purposes. Micromint reserves the right to discontinue production and change specifications and prices at any time and without notice.

This product is intended for use in normal commercial applications. Applications requiring extended temperature and unusual environmental requirements, or high reliability applications such as military, medical like-support or life-sustaining equipment, are specifically not recommended without additional processing by Micromint for such applications.

Micromint assumes no responsibility for the use of any circuitry other than circuitry embodied in a Micromint product. No other circuits, patents, or licenses are implied.

LIFE SUPPORT POLICY

Micromint's products are not authorized for use as critical components in life-support devices or systems. Life support devices or systems are devices or systems which (a) are intended for surgical implant in the body or (b) support or sustain life, and whose failure to perform, when properly used in accordance with instructions provided on the labeling, can be reasonably expected to result in significant injury to the user.

A critical component is any life-support device or system whose failure to perform can be reasonably expected to cause failure of the life-support device or system, or to affect its safety or effectiveness.

RTC-AD12E
Analog-to-Digital and Digital-to-Analog
Expansion Board

1. Specifications

- Drop-in replacement for the RTC-AD12 but also offers enhanced features.

- 12-bit plus sign analog-to-digital converter (ADC)
  - 8 Single-ended or 4 differential channels
  - Gain of each channel is programmable through software for x1, x2, x4, or x8
  - Bipolar (-5 to +5V) input range
  - Conversion time = 15.68μS (min.)
  - Conversion start to data valid time = 34.5μS (min.)
  - ADC auto-zeros itself before each conversion
  - ADC can self-calibrate at the user's request

- Two 2-channel, 12-bit digital-to-analog converters (DAC)
  - Board can be configured with 0, 2, or 4 channels
  - Outputs are jumper selected for 0-5V, 0-10V, or -5 to +5V
  - Output settling time = 10μS (max.)

2. General Description

The RTC-AD12E is an analog-to-digital and digital-to-analog expansion board for the RTC bus. The board features an 8-channel (single-ended) or 4-channel (differential) 12-bit plus sign ADC and up to four channels of 12-bit analog output.

The ADC section can be hard jumpered for a gain of 1 across all the channels or each channel's gain may be set through software. The ADC can accept voltages in the range of -5 to +5 volts. Analog signals on brought onto the board through pin header J3.

The DAC section of the board can be populated for 0, 2, or 4 output channels. The outputs can be set through jumpers to be 0-5V, 0-10V, or -5 to +5V. The analog outputs come off the board through a 2x4 pin header (J5).

3. Power Requirements

Power is supplied to the board through the RTC bus or connector J1. The board requires the following power:

+5V @ 175mA
+12V @ 50mA
-12V @ 50mA



J1 - Right-angle Power Connector

As stated previously, the power may be supplied to the RTCAD-12E through the RTC bus or from connector J1. Jumper JP1 selects which supply is used.



JP1
JP1 shown selecting power from J1

4. Addressing

The RTC-AD12E base address is set using jumper JP24. JP24 allows the base address to be set to 8000h, A000h, C000h, E000h.

JP24



JP2 shown selecting a base address of E000H

This address when combined with a function address allows you to control the various sections of the board. The function address' for the board are selected through your software and are as follows:

+00H = ADC *CS (U9 Chip select)
+10H = Input port and ADC EOC status
+20H = ADC *CAL (U9 Calibrate select)

+30H = not used

+40H = Ch. 0 (U13) LS input latch
+41H = Ch. 0 (U13) MS input latch
+42H = Ch. 1 (U13) LS input latch
+43H = Ch. 1 (U13) MS input latch
+50H = Ch. 0 & 1 (U13) Transfer data
+60H = Ch. 2 (U11) LS input latch
+61H = Ch. 2 (U11) MS input latch
+62H = Ch. 3 (U11) LS input latch
+63H = Ch. 3 (U11) MS input latch
+70H = Ch. 2 & 3 (U11) Transfer data

To use a particular function you write to (or read from) the combined address of Base + Function. For example, to start the ADC's self-calibration function you would write to the base address (E000H) + the function (20H) = E020H.

Note: The actual address E020H above includes the following ranges:

E020H-E02FH
E120H-E12FH
E220H-E22FH
.
.
.
FE20H-FE2FH
FF20H-FF2FH

Therefore any address in this range can be used to initiate the self-calibration function. For simplicity we will use only E020H in all the examples. This situation exists for each actual address and is discussed here only for reference and exposing address conflicts.

5. Using the Analog-to-Digital Converter

The ADC section of the RTC-AD12E consists of three parts, a multiplexer (U6), a programmable gain amplifier (U8), and the actual ADC (U9). Accessing the analog-to-digital converter can be as simple as two write and three read statements.

5.1 Analog Input Connector - J3

The analog signals are brought onto the board through connector J3. The drawings below show how the connector is laid out.



5.2 The A/D Conversion Process

Each time a conversion starts, the ADC automatically goes through an auto-zero cycle to minimize zero errors. The ADC can also be put into an auto-calibrate cycle by pulling its *CAL pin low. The auto-calibrate calibrate cycle will correct zero, full-scale, and linearity errors. To start the auto-calibrate cycle a dummy value is written to the address equal to Base + 20H. For example:

```
XBY(0E020H) = 0   (in BASIC-52)

OUT $E020,0   (in BASIC-180)

POKE($E020,0)   (in BASIC-11)
```

To read an analog input the channel number and gain must be sent to the address equal to Base + 00H. The control word which defines the channel number and gain is configured as follows:

ADC Control Word **Address = Base + 00H**

Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0

Bit 7	Bit 6	Gain
0	0	1
0	1	2
1	0	4
1	1	8

Bit 2	Bit 1	Bit 0	Ch
0	0	0	0
0	0	1	1
0	0	1	0
0	1	0	2
0	1	1	3
1	0	0	4
1	0	1	5
1	1	0	6
1	1	1	7

Note that bits 3, 4, & 5 are not used. From the tables above we can see that if we want to read channel 0 with a gain of 1 the control word would be 00H. If we wish to read channel 6 with a gain of 4, the control word would be 86H.

The End of Conversion (EOC) status bit is used to detect when a conversion is complete. The EOC status bit is obtained by reading the address, Base + 10H (E010H in this example). **The EOC bit will be low (logical 0) during a conversion or when the analog-to-digital converter is in its calibration cycle. The bit will be high (logical 1) when a conversion is completed.**

EOC Status **Address = Base + 10H**

Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
x	x	x	x	x	x	x	EOC

To determine the status of the conversion, a dummy value is first written to the EOC bit. The conversion is then started and the EOC bit is read. Once the EOC bit goes high the conversion is completed.

After it has been determined that the conversion is complete, the data may be read. This is accomplished by doing two reads of the analog-to-digital converter at address Base + 00H. The first read produces the four most significant bits and the sign bit. The second read produces the eight least significant bits.

First Read - MSBs & sign **Address = Base + 00H**

Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
Sign	Sign	Sign	Sign	Bit 11	Bit 10	Bit 9	Bit 8

Second Read - LSBs **Address = Base + 00H**

Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0

Reading the ADC can be broken down into four steps.

1 - Reset the EOC status by writing a dummy value to it.

2 - Send a control word containing the gain and channel number to start the conversion.

3 - Wait for EOC.

4 - Read the data from the ADC.

5.3 Sample A/D Programs

BASIC-52 ADC sample

```
10 REM BASIC-52 - RTC-AD12E ADC sample program
20 REM ***** Address Initialization *****
30 ADCB = 0E000h
40 ADCC = 0E020h
50 ADCE = 0E010h
60 REM ***** Calibrate the ADC *****
70 XBY(ADCE)=0
80 XBY(ADCC) = 0 : REM start by pulsing *CAL low
90 IF ((XBY(ADCE).AND.1)=0) THEN 90
100 REM ***** Choose a channel, set the gain, and read the channel *****
110 INPUT "Select a channel (0-7)",CHNL
120 INPUT "Select a gain (1,2,4,8)",GN
130 IF GN=1 THEN GN=0H
140 IF GN=2 THEN GN=40H
150 IF GN=4 THEN GN=80H
160 IF GN=8 THEN GN=0C0H
170 XBY(ADCE)=0
```

```
180 XBY(ADCB) = (GN + CHNL) : REM start the conversion
190 IF ((XBY(ADCE).AND.1)=0) THEN 190
200 ADCH = XBY(ADCB)
205 ADCH=ADCH.AND.1Fh
210 ADCL = XBY(ADCB)
220 IF (ADCH<16) THEN ADC=((256*(ADCH.AND.15)+ADCL)/4096)*5
230 IF (ADCH>15) THEN ADC=-(((1FFFH.XOR.(256*ADCH+ADCL))+1)/4096)*5
240 PRINT "Channel ",CHNL, " reads ",ADC, " volts"
250 GOTO 110
```

BASIC-180 ADC sample

```
10 'BASIC-180 - RTC-AD12E ADC sample program
20 INTEGER ADCB, ADCE, ADCC, ADCH, ADCL, GN, CHNL, EOC
30 REAL ADC
40 PRINT
50 ' ***** Address Initialization *****
60 ADCB = $E000
70 ADCE = $E010
80 ADCC = $E020
90 ' ***** Calibrate the ADC *****
95 OUT ADCE,0
100 OUT ADCC,0     'start the auto-calibrate cycle
110 IF BAND(INP(ADCE),1)=0 THEN 110
120 ' ***** Choose a channel, set the gain, and read it *****
130 PRINT "Select a channel (0-7)"
140 INPUT CHNL
150 PRINT "Select a gain (1,2,4,8)"
160 INPUT GN
170 IF GN=1 THEN GN=$0
180 IF GN=2 THEN GN=$40
190 IF GN=4 THEN GN=$80
200 IF GN=8 THEN GN=$C0
205 OUT ADCE,0
210 OUT ADCB,(GN+CHNL)      'start the conversion
220 IF BAND((INP(ADCE)),1)=0 THEN 220     'wait for EOC to go low
230 ADCH=INP(ADCB)      'read MSBs and sign
240 ADCL=INP(ADCB)      'read LSBs
250 'determine if value is pos. or neg. and convert
260 IF ADCH<16 THEN ADC=((256*(BAND(ADCH,15))+ADCL)/4096)*5
270 IF ADCH>15 THEN ADC=-(((BXOR($FFFF,(256*ADCH+ADCL))+1)/4096)*5)
280 PRINT "Channel ";CHNL;" reads ";ADC;" volts"
290 GOTO 130
```

BASIC-11 ADC sample

```
10 REM BASIC-11 - RTC-AD12E sample program
20 REM ***** Address Initialization *****
30 B=$C000
40 C=$C020
50 E=$C010
60 REM ***** Calibrate the ADC *****
70 POKE(E,$0)
80 POKE(C,$0) : REM start by pulsing *CAL low
90 IF ((PEEK(E).AND.1)=0) THEN 90
100 REM ***** Choose a channel, set the gain, and read the channel *****
110 INPUT "Select a channel (0-7)",CH
120 INPUT "Select a gain (1,2,4,8)",GN
130 IF GN=1 THEN 140 ELSE 150
140 GN=$0
150 IF GN=2 THEN 160 ELSE 170
160 GN=$40
170 IF GN=4 THEN 180 ELSE 190
180 GN=$80
190 IF GN=8 THEN 200 ELSE 210
200 GN=$C0
210 POKE(E,$0)
220 POKE(B,(GN+CH)) : REM start the conversion
230 IF ((PEEK(E).AND.1)=0) THEN 230
240 H=PEEK(B)
250 H=H.AND.$1F
260 L=PEEK(B)
280 IF (H<16) THEN 290 ELSE 320
290 AD=(256*(H.AND.15)+L)
300 N=1
310 GOTO 360
320 IF (H>15) THEN 330 ELSE 430
330 H=H.AND.$0F
340 AD=($FFF.EOR.(256*H+L))+1
350 N=-1
360 A1=AD/819 : REM 819 = 1V
370 AX=(AD-(819*A1))*10
380 A2=AX/819
390 AX=(AX-(819*A2))*10
400 A3=AX/819
410 AX=(AX-(819*A3))*10
420 A4=AX/819
430 A1=A1*N
440 PRINT "Channel ";CH;" reads ";A1;".";A2;A3;A4
450 PRINT
460 GOTO 110
```

5.4 Differential versus Single-ended Inputs

The previous discussion and sample code have dealt with using the RTC-AD12E in single-ended mode. Using the board in differential mode is very similar. In single-ended mode the signal being measured is referenced to the ground of the RTC-AD12E. In differential mode the signal is not referenced to ground.

Single-Ended Mode



Differential Mode



JP3 is used to select single-ended or differential mode as shown above. No changes in software are needed except to note that in differential mode there are only four channels instead of eight. The four differential channels are read as channels 0-3.

5.5 Emulating the RTCAD-12

The RTC-AD12E can also be used as a drop-in replacement for the original RTC-AD12 board. To do this, JP5 is set on pins 3-5 and 4-6 so that the gain applied to the inputs will always be x1.



JP5 shown selecting RTC-AD12E mode

6. Using the Digital-to-Analog Converters

Depending on which version of the RTC-AD12E you have, the board may contain 0, 2, or 4 channels of digital-to-analog conversion. The two DACs, U11 and U13, each provide two output channels. The DAC outputs are connected to J5.

6.1 Analog Output Connector J5



The output of each channel can be set to one of three ranges: 0 to +5V, 0 to +10V, or -5 to +5V. Selection is accomplished through jumpers JP6 (Ch. 0), JP7 (Ch. 1), JP8 (Ch. 2), and JP9 (Ch. 3). **Please refer to the silkscreen for the location and orientation of these jumpers.**



The 0 to +10V output mode is a special case. The maximum output of a channel configured in this mode is approximately 80% of the +12V supply. For example, if your +12V supply is actually supplying 11.8 volts then the maximum output of the DAC will be approximately 9.44 volts. This anomaly does not change the relationship between the data input to the DAC and the output. Sending 0 to the channel will still result in a 0V output and sending 2048 (800h) will still result in a +5V output. The only difference that will be noted is the maximum voltage which can be output.

The RTC-AD12E can however use ±15 volt supplies instead of the normal ±12 volt system supplies. The ±15 volt lines are connected to power connector J1. **When doing this the jumpers must be removed from JP1 in order to disconnect these power lines from the RTC bus. Connecting a ±15 volt power supply to the board with the JP1 jumpers installed can damage other boards in the RTC stack!**

6.2 The D/A Conversion Process

Control of the DAC is accomplished by writing a value to the input latches and then transferring the data to the output. Each channel has two input latches associated with it, the 4-bit most significant and the 8-bit least significant. Either of these latches may be written to first. Writing to the latches is quite simple. For example, if channel 1 is configured for the -5 to +5V output range and we wish to have a 4V output then the following must be done.

10V (output range) ÷ 4096 = 0.00244V (resolution)

Since the range is -5 to +5V then 0 = 2048 = 800h. So if the value to be output is greater than or equal to zero then we must add 2048 (or 800h) to it.

4V ÷ 0.00244V = 1639 = 667h

1639 + 2048 = 3687 = value sent to DAC
or
667h + 800h = E67h = value sent to DAC

MS 4 bits	LS 8 bits
E	67

With this information we are now ready to write the data to the DAC. Using our example board address of E000h:

Write 0Eh to E042h	Load MS 4 bits
Write 67h to E043h	Load LS 8 bits
Write a dummy value to E050h	Transfer the data to the output

It is important to note that when a transfer is begun, the data for both channels is transferred from the input latches to the outputs at the same time. Because of this you can load the input latches for both channels and then do a transfer. This will update both outputs simultaneously.

6.3 Sample D/A Programs

BASIC-52 DAC Sample

```
10 REM BASIC-52 - RTC-AD12E DAC sample
20 PRINT
30 REM ***** addresses *****
40 BASE = 0E000h
50 T01 = 50h
60 T23 = 70h
70 REM ***** select a channel *****
80 PRINT
90 INPUT "Select a channel (0,1,2,3) ",CH
100 IF CH=0 THEN CH=40H
110 IF CH=1 THEN CH=42H
120 IF CH=2 THEN CH=60H
130 IF CH=3 THEN CH=62H
140 PRINT
150 REM ***** select an output range *****
160 PRINT "What output range is selected by JP6/7/8/9?"
170 PRINT " 1  -  -5 to +5V"
180 PRINT " 2  -  0 to +5V"
190 PRINT " 3  -  0 to +10V"
200 INPUT "Enter your choice (1,2,3) ",RG
210 PRINT
220 REM ***** select a voltage to output *****
230 INPUT "Enter the voltage you wish to output ",V
240 PRINT
250 REM ***** convert and output *****
260 PRINT "Converting the data and setting the output ..."
270 IF RG=2 THEN  GOTO 400
280 IF RG=3 THEN  GOTO 470
290 IF (V<0) THEN  GOTO 320
300 V = (INT(V/0.0024415)) + 2048
310 GOTO 340
320 V = (5 + V)
330 V = (V/0.00244)
340 MSB = (V.AND.0F00h)/0FFh
350 LSB = V.AND.0FFh
360 XBY(BASE+CH) = LSB
370 XBY(BASE+CH+1) = MSB
380 GOSUB 540
390 GOTO 80
400 V = (INT(V/0.0012207)
410 MSB = (V.AND.0F00h)/0FFh
420 LSB = V.AND.0FFh
430 XBY(BASE+CH) = LSB
440 XBY(BASE+CH+1) = MSB
450 GOSUB 540
460 GOTO 80
470 V = INT(V/0.002442)
480 MSB = (V.AND.0F00h)/0FFh
490 LSB = V.AND.0FFh
```

```
500 XBY(BASE+CH) = LSB
510 XBY(BASE+CH+1) = MSB
520 GOSUB 540
530 GOTO 80
540 IF (CH<60H) THEN XBY(BASE+T01)=0
550 IF (CH>42H) THEN XBY(BASE+T23)=0
560 RETURN
```

BASIC-180 DAC Sample

```
10 'BASIC-180 - RTC-AD12E DAC sample program
20 INTEGER BASE, T01, T23, CH, RG, MSB, LSB, VT
30 REAL V
40 PRINT
50 ' ***** Address Initialization *****
60 BASE = $E000
70 T01 = $50
80 T23 = $70
90 ' ***** Select a Channel *****
95 PRINT
100 PRINT "Select a channel (0,1,2,3) "
110 INPUT CH
120 IF CH=0 THEN CH=$40
130 IF CH=1 THEN CH=$42
140 IF CH=2 THEN CH=$60
150 IF CH=3 THEN CH=$62
160 PRINT
170 ' ***** Select an Output Range *****
180 PRINT "What output range is selected by JP6/7/8/9 ?"
190 PRINT " 1  -  -5 to +5V"
200 PRINT " 2  -  0 to +5V"
210 PRINT " 3  -  0 to +10V"
220 PRINT "Enter your choice (1,2,3) "
230 INPUT RG
240 PRINT
250 PRINT "Enter the voltage you wish to output "
260 INPUT v
270 PRINT
280 ' ***** Convert Data *****
290 PRINT "Converting the data and setting the output ..."
300 IF RG=2 THEN GOTO 420     'range = 0 to +5
310 IF RG=3 THEN GOTO 490     'range = 0 to +10
315 ' ***** Output range = -5 to +5 *****
320 IF (V<0) THEN GOTO 350
330 VT = (V/0.0024415) + 2048
340 GOTO 370
350 V = 5 + V
360 VT = V/0.0024415
370 MSB = BAND(VT,$F00)/$FF
380 LSB = BAND(VT,$FF)
390 OUT (BASE+CH),LSB
400 OUT (BASE+CH+1),MSB
410 GOTO 540
```

```
415 ' ***** Output range = 0 to +5 *****
420 VT = V/0.0012207
430 MSB = BAND(VT,$F00)/$FF
440 LSB = BAND(VT,$FF)
450 OUT (BASE+CH),LSB
460 OUT (BASE+CH+1),MSB
470 GOTO 540
480 ' ***** Output range = 0 to +10 *****
490 VT = V/0.0024415
500 MSB = BAND(VT,$F00)/$FF
510 LSB = BAND(VT,$FF)
520 OUT (BASE+CH),LSB
530 OUT (BASE+CH+1),MSB
535 ' ***** Transfer data *****
540 IF (CH<$60) THEN OUT (BASE+T01),0
550 IF (CH>$42) THEN OUT (BASE+T23),0
560 GOTO 95
```

BASIC-11 DAC Sample

```
10 REM BASIC-11 DAC sample
20 PRINT
30 REM ***** addresses *****
40 B=$C000
50 T1=$50
60 T2=$70
70 REM ***** select a channel *****
80 PRINT
90 INPUT "Select a channel (0,1,2,3) ",CH
100 IF CH=0 THEN 110 ELSE 130
110 CH=$40
120 GOTO 210
130 IF CH=1 THEN 140 ELSE 160
140 CH=$42
150 GOTO 210
160 IF CH=2 THEN 170 ELSE 190
170 CH=$60
180 GOTO 210
190 IF CH=3 THEN 200 ELSE 210
200 CH=$62
210 PRINT
220 REM ***** select an output range *****
230 PRINT "What output range is selected by JP6/7/8/9?"
240 PRINT " 1  -  -5 to +5V"
250 PRINT " 2  -  0 to +5V"
260 PRINT " 3  -  0 to +10V"
270 INPUT "Enter your choice (1,2,3) ",RG
280 PRINT
290 REM ***** select a voltage to output *****
300 PRINT "Enter the voltage you wish to output in the following"
310 PRINT "format:  I1 I2 . I3 I4 I5 " : PRINT
320 PRINT "ie: -1.5V - I1=0, I2=-1, I3=5, I4=0, I5=0" : PRINT
330 INPUT "Enter the value for I1: ",I1
```

```
340 INPUT "Enter the value for I2: ",I2
350 INPUT "Enter the value for I3: ",I3
360 INPUT "Enter the value for I4: ",I4
370 INPUT "Enter the value for I5: ",I5
380 PRINT
390 REM ***** convert and output *****
400 PRINT "Converting the data and setting the output ..."
410 IF RG=2 THEN 610
420 IF RG=3 THEN 710
430 IF I2<0 THEN 490
440 V=2048+(I2*409)
450 V=V+(I3*41)
460 V=V+(I4*4)
470 V=V+(I5*1)
480 GOTO 550
490 I2=-1*I2
500 V=I2*409
510 V=V+(I3*41)
520 V=V+(I4*4)
530 V=V+(I5*1)
540 V=2048-V
550 MS=(V.AND.$F00)/$FF
560 LS=V.AND.$FF
570 POKE(B+CH,LS)
580 POKE(B+CH+1,MS)
590 GOSUB 820
600 GOTO 80
610 V=I2*819
620 V=V+(I3*82)
630 V=V+(I4*8)
640 V=V+(I5*1)
650 MS=(V.AND.$F00)/$FF
660 LS=V.AND.$FF
670 POKE(B+CH,LS)
680 POKE(B+CH+1,MS)
690 GOSUB 820
700 GOTO 80
710 V=I1*4095
720 V=V+(I2*409)
730 V=V+(I3*41)
740 V=V+(I4*4)
750 V=V+(I5*1)
760 MS=(V.AND.$F00)/$FF
770 LS=V.AND.$FF
780 POKE(B+CH,LS)
790 POKE(B+CH+1,MS)
800 GOSUB 820
810 GOTO 80
820 IF (CH<$60) THEN 830 ELSE 840
830 POKE(B+T1,0)
840 IF (CH>$42) THEN 850 ELSE 860
850 POKE(B+T2,0)
860 RETURN
```

7. Adjustment and Calibration

All RTC-AD12E adjustments are calibrated at the factory. If you feel a unit is in need of adjustment, we recommend the unit be sent back for calibration. A calibration charge of $20.00 must accompany all returned units. This charge is only for calibration and does not cover the replacement or repair of any components that have been damaged through misuse.

The accuracy of the RTC-AD12E depends on the accuracy of the equipment used in calibration. Do not attempt calibration without the proper equipment. A 4 1/2 digit multimeter is required for proper calibration of the RTC-AD12E.

7.1 Reference Voltage Calibration

To calibrate the reference voltage, set the multimeter to a range which will measure a +5 volt signal. Connect the positive lead of the meter to test point TP1 on the board. Connect the common (ground, negative) lead to test point TP2. Once the test leads are connected adjust potentiometer P1 for a reading of +5.000 volts.

7.2 Offset Calibration

The RTC-AD12E has two adjustments for offset. Potentiomer P3 adjusts the input offset and potentiometer P2 adjusts the output offset of the ADC section. To calibrate the offset a program must be run to continuously read one of the input channels at the maximum gain and then at the minimum gain.

To begin the alignment, make sure the board is in the single-ended mode and short Input 0 to ground. Make sure that the shorting wire is as short as possible. Set the multimeter to it's lowest voltage range. Connect the common (ground, negative) lead of the meter to pin 1 of J3 (ground). Connect the positive lead to pin 1 of U16 (ADC1251).

Once the test program is running, it should be reading Input 0 at the highest gain (8). Adjust the input offset (P3) for a reading as close to 0.000V as possible. Next instruct the program to begin reading Input 0 at the lowest gain (1). Now adjust the output offset (P2) until a reading as close to 0.000V is again reached. These two steps should be repeated until the reading does not change, or changes very little, when the gain is switched between the maximum and minimum.

8. RTC-AD12E Silkscreen



9. RTC-AD12E Schematic




MICROMINT, INC.
4 PARK STREET
VERNON, CT 06066
RTC 12-BIT ANALOG I/O
JEFF BACHIOCHI
REV 1.0
SHEET 2 OF 3
DATE: 12/7/94
Data Bus
Control Bus


MICROMINT, INC.
4 PARK STREET
VERNON, CT 06066
RTC 12-BIT ANALOG I/O
JEFF BACHIOCHI
12/7/94
REV 1.0
SHEET 3 OF 3
PGA205 Supply Voltages
WR*(CSD2+0)=LSB Vout1
WR*(CSD2+1)=MSB Vout1
WR*(CSD2+2)=LSB Vout2
WR*(CSD2+3)=MSB Vout2
WR*CSD3=XFER all Values
WR*(CSD0+0)=LSB Vout1
WR*(CSD0+1)=MSB Vout1
WR*(CSD0+2)=LSB Vout2
WR*(CSD0+3)=MSB Vout2
WR*CSD1=XFER all Values
ADJUST TO 5.000V
U11 AD7237
U13 AD7237
REF02 U3
Control Bus
Data Bus
Address Bus

10. RTC-AD12E Parts List

Printed Circuit Board

PCB1	RTC-AD12E PCB

Integrated Circuits

U1	74LS138
U2	74LS04
U3	REF02CP
U4	74HCTLS02
U5	74LS373
U6	HI3-0518-8
U7	74LS244
U8	PGA205
U9	ADC1251
U10	74LS74
U11, U13	AD 7237
U12	LM29021

Resistors

R1, R3	180Ω, 1/4W, 1%
R2, R4	576Ω, 1/4W, 1%
R5, R6	27kΩ, 1/4W, 5%
R7, R8	680Ω, 1/4W, 5%
R9	470Ω, 1/4W, 5%
R10, R11	1kΩ, 1/4W, 5%
R12	20kΩ, 1/4W, 5%
R13, R14	10kΩ, 1/4W, 5%
R15	15kΩ, 1/4W, 5%
R16	6.2kΩ, 1/4W, 5%
SIP1	10kΩ, 9 element, 10 pin
P1	10kΩ multi-turn pot., side adjust
P2	50kΩ multi-turn pot., side adjust
P3	200kΩ multi-turn pot., side adjust

Capacitors

C1-C8, C10-C23	0.1μF, 50V, .10", Monolythic
C9, C25-C32, C34, C35	10μF, 16V, Tantalum
C33	0.01μF
C36	470pF, 1/8", Ceramic

Connectors

J1	1x4 Header, right angle
J3	2x8 Header
J5, JP24	2x4 Header
JP3, JP4	1x3 Header
JP3, JP5-JP9	2x3 Header
JP24	2x2 Header
JP15, JP16	2x13 Header, VSC63, Long

Sockets

SK1, SK8	16 pin DIP socket
SK2, SK4, SK10, SK12	14 pin DIP socket
SK3	8 pin DIP socket
SK5, SK7,	20 pin DIP socket
SK6	18 pin DIP socket
SK9	24 pin DIP socket
SK11, SK13	24 pin DIP socket, 0.300"

Semiconductors

Q1	LM337LZ
Q2	LM317LZ